ublic abstract aspect ObserverProtocol {  

    /**

     * This interface is used by extending aspects to say what types

     * can be <i>Subject</i>s. It models the <i>Subject</i> role.

     */

    protected interface Subject  { }    

    /**

     * This interface is used by extending aspects to say what types

     * can be <i>Observer</i>s. It models the <i>Observer</i> role.

     */

    protected interface Observer { }

    /**

     * Stores the mapping between <i>Subject</i>s and <i>

     * Observer</i>s. For each <i>Subject</i>, a <code>LinkedList</code>

     * is of its <i>Observer</i>s is stored.

     */

    private WeakHashMap perSubjectObservers;

    /**

     * Returns a <code>Collection</code> of the <i>Observer</i>s of 

     * a particular subject. Used internally.

     *

     * @param subject the <i>subject</i> for which to return the <i>Observer</i>s

     * @return a <code>Collection</code> of s's <i>Observer</i>s

     */

    protected List getObservers(Subject subject) { 

        if (perSubjectObservers == null) {

            perSubjectObservers = new WeakHashMap();

        }

        List observers = (List)perSubjectObservers.get(subject);

        if ( observers == null ) {

            observers = new LinkedList();

            perSubjectObservers.put(subject, observers);

        }

        return observers;

    }

    /**

     * Adds an <i>Observer</i> to a <i>Subject</i>. This is the equivalent of <i>

     * attach()</i>, but is a method on the pattern aspect, not the 

     * <i>Subject</i>. 

     *

     * @param s the <i>Subject</i> to attach a new <i>Observer</i> to

     * @param o the new <i>Observer</i> to attach

     */ 

    public void    addObserver(Subject subject, Observer observer) { 

        getObservers(subject).add(observer);    

    }

    /**

     * Removes an observer from a <i>Subject</i>. This is the equivalent of <i>

     * detach()</i>, but is a method on the pattern aspect, not the <i>Subject</i>. 

     *

     * @param s the <i>Subject</i> to remove the <i>Observer</i> from

     * @param o the <i>Observer</i> to remove

     */ 

    public void removeObserver(Subject subject, Observer observer) { 

        getObservers(subject).remove(observer); 

    }

    /**

     * The join points after which to do the update.

     * It replaces the normally scattered calls to <i>notify()</i>. To be

     * concretized by sub-aspects.

     */ 

    protected abstract pointcut subjectChange(Subject s);

    /**

     * Calls <code>updateObserver(..)</code> after a change of interest to 

     * update each <i>Observer</i>.

     *

     * @param subject the <i>Subject</i> on which the change occured

     */

    after(Subject subject): subjectChange(subject) {

        Iterator iter = getObservers(subject).iterator();

        while ( iter.hasNext() ) {

            updateObserver(subject, ((Observer)iter.next()));

        }

    } 

   /**

     * Defines how each <i>Observer</i> is to be updated when a change

     * to a <i>Subject</i> occurs. To be concretized by sub-aspects.

     *

     * @param subject the <i>Subject</i> on which a change of interest occured

     * @param observer the <i>Observer</i> to be notifed of the change  

     */

    protected abstract void updateObserver(Subject subject, Observer observer);

}
public aspect UpdateStateObserver extends ObserverProtocol {

declare parents: IFacade implements Observer;

declare parents: Complaint ||

 



 Employee ||  

 



 Symptom ||

 



 MedicalSpeciality ||

 



 HealthUnit implements Subject;

after(Subject s) : call(* CommandRequest+.put(String, Object)) && this(CommandServlet) &&
args(.., s) {


addObserver(s, HealthWatcherFacade.getInstance());

}

protected pointcut subjectChange(Subject subject): 


this(CommandServlet+) && call(* Subject+.set*(..)) && target(subject);

protected void updateObserver(Subject subject, Observer observer) {


if(observer instanceof IFacade) {



IFacade facade= (IFacade) observer;



if(subject instanceof Employee){




facade.updateEmployee((Employee) subject);



} else if(subject instanceof Complaint){




facade.updateComplaint((Complaint) subject);



} else if(subject instanceof HealthUnit){




facade.updateHealthUnit((HealthUnit) subject);



} else if(subject instanceof Symptom){




facade.updateSymptom((Symptom) subject);



} else if(subject instanceof MedicalSpeciality){




facade.updateMedicalSpeciality((MedicalSpeciality) subject);



}


}

}
}
